
  mba77a@yahoo.com مھند عبدالحفيظ –ي خمسة أيام  فLINQ to SQLتعلم 

  
 رغم انه توجد مواضيع عديدة غنية عن ھذه التقنية LINQ) نسبيا(ھديتي المتواضعة لكم ھي عن القادم الجديد 

واحب ان اذكركم بأني مجرد ھاو للبرمجة تتواضع معرفتي ومعلوماتي امام الاساتذة .فلامانع من التنوع 
فلا أريد . حوا خطأي  وسأكون لكم ممنوناوالمحترفين و الاعضاء في ھذا المنتدى فإن اخطأت أرجو منكم أن تصح

ويترتب على كوني ھاو ايضا فترات من الانقطاع المفاجئ نتيجة .أن أعرض معلومات خاطئة لأخوتي العرب 
  .ظروف عملي 

  
 وانا انشرھا بالتزامن في منتدى اخر LINQ to SQLسأضع سلسة من المقالات بلغة سھلة مبسطة عن تقنية 

  على ارض الواقع  LINQوالمقالات تحتوي ارشادات عملية لبناء برنامج لتطبيق .. النفع صديق من اجل ان يعم 
لأن ادوات انشائھا وتطويرھا مضمنة ومتوفرة   SQL Server Compactوقد اخترت التعامل مع قواعد بيانات 

ليھا ينطبق على نسختي وما ينطبق ع.   مباشرةLINQبالاضافة لدعمھا لتقنية   ٢٠٠٨لكل مستخدم للفيجوال بيزيك 
 Express و Enterprise.   

  
  ...نبدأ باسم الله 

  :مقدمة 
  
  

  LINQمقدمة عن 
  

   ؟LINQما ھي 
  

نات بكافة اشكالھا واينما كانت  ھي مجموعة تعليمات جديدة للتعامل مع البياLINQباختصار 
)List,DataSet,DataBase,Array…etc  (  وھي شديدة الشبه بSQLولكن الفرق أن  جيدا  التي نعرفھاSQL تتعامل 

 و Collectionsكالمجموعات ( تتعامل مع كائنات فيجوال بيزيك LINQمع كائنات قاعدة البينات كالجداول مثلا مباشرة بينما 
 ) ... Engine( تترجم وبالتالي يتم تدقيقھا على مستوى قاعدة البينات SQLعبارات : بعبارة أخرى  ) Arraysالمصفوفات 

   .LINQوھذا احد جوانب قوة  . Visual Basic فھي تترجم وتدقق على مستوى LINQعبارات اما 
  

   :LINQ إلى DAOمن 
ثم بھرتنا .. المملة والبطيئة DAOفقد بدأنا قديما مع .  الان في قمة الھرم LINQوفي مسيرة تطور التعامل مع البيانات فإن 

ADO بنفس الكائنات مستفيدة من مع معظم مصادر البيانات بكائناتھا المتخصصة وقدرتھا على التعامل OLE DB ..  ثم
ADO.NET التي تبنت منصة  .NETودعم  للتطوير OOP   واخيرا تم تغليفADO.NET بالمزود  LINQ to DataSet 

نات  لتشمل كافة انواع البياLINQوتوسيع   SQL Server للتعامل المباشر مع قواعد بيانات  LINQ to SQLو اتاحة 
)objects و XML و SQL ( كما سنرى فھي البديل الأمثل للتعامل مع البيانات.  
 

سنجد أن  ..RDOو  DAO ظل كثير من المبرمجين متمسكين ب ADOوكما سبق ان حدث عندما قدمت لنا مايكروسوفت 
 لا LINQاسي الشديد ل  ولاضير في ھذا فانا رغم حمADO.net ويفضلون LINQيرا من المبرمجين لازالوا يحجمون عن كث

  . ستفرض نفسھا بقوةLINQولكن اؤكد لكم  .ADOازال الجأ احيانا لكائنات 
  

   ؟LINQلماذا اترك ما تعلمته وجربته واتجه لما يسمى 
  visual basic.net الاولى ھي     نعم...!!!  فانت بحاجة الى تعلم لغتي برمجة LINQحسنا اذا لم تكن من مستخدمي 

نتيجة غير (جب أن تعرف قواعد الكتابة كل منھما و اخطاء كل منھما وعند حدوث خطأ منطقي ي ... SQL والثانية ھي
 و تعود لتنسخ عبارة  في قاعدة بياناتكSQLستراجع اولا برنامجك في الفيجوال ثم ستذھب لتجربة عدة عبارات ) متوقعة 

SQLبة اسم حقل فتعود  لترى اسم الحقل الصحيح وتصححه ثم  المناسبة وتضعھا في برنامجك لتكتشف أن ھناك خطأ في كتا
  تقوم بالعمل وتتمنى لو أن كلSQL لاتلبي احتياجك وتضطر لمراجعة ملفات المساعدة للبحث عن دالة SQLتكتشف ان جملة 

ن كل شيء العمل في نفس الصفحة حيث لاتضطر للمغادرة والتقصي والبحث من اجل اسم جدول او عمود أو دالة ستتمنى لو ا
 بتقديم الكائنات لك لتنقتي منھا كما اعتدنا في الفيجوال او ما يسمى بخاصية امامك كما اعتدت في بيئة الفيجوال ليقوم المحرر

IntelliSense عن ظھر  و اعمدتھا وانواع بيانات اعمدتھا يجب ان تحفظ قاعدة بياناتك وجداولھا... ولكن الواقع غير ذلك

 واذا كنت .. لتمثيل العلاقات  SQL في INNER JOINه من العلاقات فانت بحاجة لالف سطر من آآآآآآو... قلب وعلاقاتھا 
   لھذا النوع SQLتتعامل مع اكثر من نوع قواعد بيانات فستنكب على الكتب والانترنت تتعلم ميزات لغة 

  ھل ھذه حالك ؟؟؟.. ماذا بعد .... 
 والسحرة -" Wizardsالسحرة  " يستخدم ادوات الفيجوال التي تعلم الكسل والمسماة  فانت من النوع الاخر الذي...... طبعا لأ

 فانت لاتتعب كثيرا والمعالج يقوم بكل شيء فانت مستعد لتتحداني انه يمكنك كتابة برنامج كامل للتعامل مع قاعدة -دائما خبثاء 
صدقك ولكن جرب ان تنقل برنامجك لجھاز نعم ا... !!!  ووجع راسھا  LINQالبيانات بدون أي سطر كود وبدون 

اكاد اجزم انك ستحتاج للتعامل للبيانات من خلال الكود ..أو ان تكتب برنامجا حقيقيا وليس تعليميا أو تجريبيا .. اخر 
  .وستصبح من اصحاب النوع الأول ..

ن تصميم بنية قاعدة بيناتك في فيجوال وستبقى بعد الانتھاء م...سھلة بسيطة سريعة وقوية  ... LINQاستخدم ... اليك الحل 
لأن الفيجوال )   فھي تأتي بالانكليزية بمعنى ساحر Wizardsأقصد ( بيزيك ولن تحتاج لمغادرته ولا لاستخدام المشعوذين 

 سيقوم بتلقينك ما تريد كتابته وفوق ھذا سيخبرك ھو ببنية قاعدة البينات وايضا سيمثل لك العلاقات خير تمثيل حيث تصل
واما الاخطاء فانتھى عھد ان تعمل !!  ستصبح اربع نقاط Inner Joinتخيلوا اربع اسطر (.) ..للعلاقة المطلوبة بنقطة فقط 

السيد  ..LINQفالراعي الرسمي ل  ...Null وماذا اعمل ب ؟؟int32وما ھو مقابل !..  ؟؟Doubleفحص اين اضع 
IntelliSenseا اخطاء  سيقوم بالعمل ويوبخك اذا اخطأت ھكذ....  

  
  
  
  
  

  ...ساستخدم اسلوب اكثر تقليدية ... الظاھر ان اسلوبي لم يعجبك .... لم تقتنع 
  :LINQ للسيد المحترم القارئ نقدم لكم مزايا 

 Customersمجموعة  يصبح Customerجدول ( التعامل مع كائنات قواعد البينات ككائنات برمجية  -١
Collection ( التالي تفعيل خاصية أي احضارھا للبرنامج وبIntelliSense الرائعة . 

مع ان جماعة مايكروسوفت  .SQL فيجوال بيزيك الاعتيادية مع البيانات بدلا مع دوال Functionsاستخدام دوال  -٢
  .SQLاضافت دوال جديدة شديدة الشبه بدوال 

 .SQL Server او Oracle واجھة واحدة للتعامل مع البيانات اينما كانت سواء بالذاكرة او بقواعد بيانات -٣
 . ومعرفة فروقاتھا بين محركات قواعد البيانات المختلفة SQLلست بحاجة لاستخدام  -٤
 .تصيد الأخطاء في الاستعلامات يتم قبل الترجمة واثناء كتابة الكود  -٥
 . كما سترون لاحقا Sub-Collections بشكل مجموعات فرعية Foreign keysتمثيل العلاقات  -٦
 .ي جديد يثير الفضول وجربھا يا اخع بعد ھذا اعتبرھا شيء اذا لم تقتن -٧

  
  ھل سنبدأ بالكلام المفيد ؟

  . لتنتقي ما يناسبك LINQحسنا اطلت الشرح والاقناع سأشرح قليلا انواع مزودات 
LINQ to SQL:  مختصة بالتعامل مع قواعد بيانات SQL Serverلاماتك  بكافة انواعھا وباختصار اداة ترجمة قوية لاستع
واھم ما يعجبني فيھا أنني استطيع استخدام أي دالة .  ترسل من تحت الطاولة دون علمك لقاعدة البيانات  SQLإلى استعلامات 

Function في جملة Where ونحن كما ذكرت لكم سنتعامل مع ھذا المزود من خلال قاعدة بيانات .  حتى دالة من تعريفك
SQL Server Compact المضمنة في VS 2009.   

LINQ to Object :  مختصة بالتعامل مع كائنات فيجوال بيزيك مثلا استعلام لمصفوفةArray او لمجموعة Collection أو 
  .IEnumerable Interface أي شيء يدعم الواجھة Listقائمة 

LINQ to XML:  لجماعة مبرمجي الويب خاصة وبيانات XML مختصين فأنا اتعقد من حيث لا أعرف عنھا الكثير فاتركھا لل
  . يذكرني بالدوس وايامه السوداء<>منظر الأقواس 

LINQ to DataSet :  ويمكن استخدام ھذا المزود لقواعد البيانات الأخرى  غير SQL Server.  مثلاAccess أو 
Oracle أو MySqL..  ألخ ھو طريقة للالتفاف على عدم دعمLINQ  لى متناول فنحضرھا إ. المباشر لھذه المحركات

  . القادرة على احضار أي قواعد بيانات DataSet قسرا باستخدام LINQاستعلامات 
  

  )linq to entities و linq to dataset و linq to sql( تطلق على كل من  LINQ to ADO.Net وللعلم فأن
  

لكن التنويع لاختلاف بنية  البيانات في كل ولاتقلقوا فالمبدأ نفسه واذا تعلمت استعلامات أي مزود ستتعلم استعلامات البقية و
  .مزود 

  
   .SQL to LINQبحمدالله انتھت المقدمة وسأبدأ في المقالات التعليمية حيث في المقالة القادمة كيفية انشاء برنامج يدعم 

  اليوم الأول
 

LINQ to SQLبناء برنامج يدعم   
  

 وبركاتــه الله ورحمـة عليكــم السلام
 

  LINQ استخدام في للمبتدئين وذلك الياء إلى لفالأ من سأبدأ
  : البرنامج عمل خطوات

  . Windows Form Application نوع من فارغ جديد برنامج انشاء .١

  . Server Explorer طريق عن البيانات قاعدة تصميم و إضافة .٢

  SQLMetal طريق عن LINQ ل العمل فئات انشاء .٣

 (تعديل-اضافة-حذف-عرض ) لبايناتا مع للتعامل LINQ استعلامات تطبيق .٤

 
 

 كماھو اتركه Form1 وحيد نموذج لك سيظھر البرنامج انشاء وبعد ... لشرح تحتاج ولا سھلة الأولى الخطوة طبعا
 .... الثانية بالخطوة ونبدأ

 
 البيانات قاعدة وتصميم بناء

 بيانات قاعدة Local نختار ثم و Add New Item ونختار Project للقائمة نذھب : جديدة بيانات قاعدة لانشاء
 ملف سنجد  . DataSetء انشا نريد لا لأننا Cancel اضغط البيانات بقاعدة الاتصال معالج فيظھر. اسم اي ونسميھا
 لقاعدة الجداول نضيف ثم ServerExplorer لنا فيفتح عليه نضغط Solution Explorer لنافذة اضيف قد الداتا

 . الازمة اتالعلاق وننشيء البيانات
 

 قائمة من نختار حيث الشھيرة NorthWind.sdf بيانات قاعدة ساستخدم وللسھولة : موجودة بيانات قاعدة إضافة
Project ثم Add Existing Item ملف تضيف ثم NorthWind.sdf المجلد في تجدھا التي :   

 

C:\Program Files\Microsoft SQL Server Compact Edition\v3.5\Samples 
 

 . Cancel فنضغط الباينات بقاعدة الاتصال معالج وسيظھر
 مجلد في احدھما بيانات قاعدتي ينشيء الفيجوال أن حيث : المحلية البيانات قواعد لكل مھمة الخطوة ھذه الان

 اختفت انھا اجأنتف البرنامج تنفيذ اثناء بيانات باضافة نقوم فعندما وبالتالي Debug مجلد في للتنفيذ والثانية البرنامج
 اختار Solution Explorer المشروع نافذة من : بالتالي نقوم لذلك server explorer في البيانات نستعرض عندما
 . Do not Copy إلى Copy to Output Directory الخاصية عدل الخصائص وفي البيانات قاعدة ملف

 
 
 الثالثة للخطوة تمھيدا المشروع نحفظ الان

 
  LINQ مع العمل فئات انشاء

 ؟ LINQ مع العمل فئات ھي ما
 داخل من بسھولة معھا التعامل يمكن كائنات شكل على واعمدتھا جداولھا بكافة البيانات قاعدة تمثل فئات مجموعة ھي

 ٣

 جميع مع اعتدنا كما وسھل واضح بشكل للمستخدم وتقدمھا البيانات قاعدة كائنات بتغليف تقوم باختصار فھي البرنامج
 . البيانات مع للتعامل اللازم الكود وتقليل بالبرنامج الاخطاء لتقليل يؤدي وھذا Net. ئاتف

 
 
 قواعد دعم ماايكروسوفت تضف لم لماذا اعلم لا ولكن O/R Designer ھي العمل فئات لانشاء المثالية البيئة ان

 : الفيجوال مع يأتي وھو SQLMetal.exe وھو البديل البرنامج سنستخدم وبالتالي ؟!!له SQL CE بيانات
  

  C:\Program Files\Microsoft SDKs\Windows\v6.0A\bin للمجلد ونذھب البرنامج نغلق اولا .١

  . Northwind.sdf ملف فيه الذي البرنامج مجلد إلى SQLMetal.exe الملف ننسخ ثم .٢

  نسخناه الذي SQLMetal.exe لملف جديد اختصار وننشئ برنامجنا مجلد في نبقى .٣

 .٤ :ةالتالي الاوامر البرنامج اسم بعد نضع ) الھدفTarget  ( خانة وفي انشأناه الذي الاختصار خصائص نفتح

/namespace:NorthWind /code:NorthWind.vb /Language:vb /pluralize NorthWind.sdf  

   : كالتالي الھدف خانة تصبح أكثر للتوضيح

"C:\Users\Muhannad\Documents\Visual Studio 
2008\Projects\SampleLINQ\SampleLINQ\SqlMetal.exe" /namespace:NorthWind 
/code:NorthWind.vb /Language:vb /pluralize NorthWind.sdf  

 

 ملف لدينا أنه وسنجد تختفي ثم للحظات الاوامر موجه نافذة ستظھر الاختصار ملف بتشغيل نقوم والان .٥
  . المطلوب وھو NorthWind.vb اسمه المجلد في جديد

 البرنامج لمجلد نذھب و Add existing Item الملف اضافة Project قائمة من ونختار مشروعنا نفتح .٦
  . NorthWind.VB الجديد ملفنا ونضيف

 يعرف لا لم الطريقة.. System.Data.Linq ل للبرنامج refrence اضافة يتطلب حيث أخيرة خطوة بقي .٧
 أضفھا System.Data.Linq ستجد NET. علامة تحت ثم Add refrence ثم Project قائمة من

 . للمشروع

 
 أو جديد جدول اضافة مثلا البيانات قاعدة تصميم في تغيير أي عند الفئات ملف انشاء اعادة يجب : ھامة ملاحظة
 في مفتوحة لبياناتا قاعدة لاتكون أن وانتبه السابق الاختصار لتشغيل الا لاتحتاج وعندھا موجود لجدول جديدة أعمدة
  . مثلا كالفيجوال اخر مكان

 

: وقد أضفت البرنامج الذي تم انشاءه في ھذا اليوم  بشكل رابط لمن يريد اختصار الوقت والتطبيق مباشرة 
LinqSample.  

 
 
 وھي الرابعة الھامة بالخطوة قريبا وسأبدأ برنامجنا في  LINQ To SQL  لاستخدام جاھزين اصبحنا الان

  . الله شاء ان LINQ تعبيرات قوة و سھولة مدى لكم ستكشف التي وھي التطبيقات
 
   التحية مع

 ٤

http://vb4arab.com/vb/uploaded/56348/11235067926.zip

اليوم الثاني   
   LINQالبدء في استخدام 

  

 باستخدام Linqأنشأنا فئات عمل  NorthWind.sdfفي الموضوع السابق انشأنا برنامجا يرتبط بقاعدة بيانات 
SQLMetal   

ولمن يريد الاستفادة من الدرس يفضل انشاء البرنامج حسب الدرس السابق لأنني سأمزج بين المعلومات النظرية 
  .والتطبيق العملي مباشرة في كافة المراحل 

  

  

   . SQLMetalباستخدام  بعد انتھينا من بناء الفئات LINQالان ساتطرق للموضوع المھم وھو التعامل مع 

ان الفئات التي تشكلت لدينا ھي صلتنا بقاعدة البيانات فبعد الان اصبح كل سجل بقاعدة اليينات ممثل بكائن وكل 
 الان نصل اليه عن طريق Customers مثلا الجدول -  أو بنية تشبھھا بالاحرى -  Listمجموعة سجلات ممثلة ب 

 وستجد لكل فئة SQLMetal التي انشأھا لنا Customerائنات من نوع  والذي يحتوي على كCustomersالكائن 
 وعند الشرح العملي ستتضح الصورة Customer خصائص ھي عبارة عن أعمدة الجدول Customerمن نوع 

  . أكثر 

  

 على كامل الفورم حيث سنستخدمھا لعرض نواتج تجاربنا مع DataGridViewالان في البرنامج ضع اداة 
   . Linq استعلامات

  

 والذي مھمته تأمين الاتصال بقاعدة DataContext ھو انشاء الكائن LINQ TO SQLالخطوة الأولى لاستخدام 
 والتي SQLMetal التي انشأھا لنا Northwindالبيانات وطبعا نحن لن ننشئ ھذا الكائن مباشرة انما ھناك الفئة 

تحت .  في الفورم Field من ھذه الفئة كمتغير محلي  وبالتالي سننشئ كائنDataContextھي اصلا مشتقة من 
   . Public Classكلمة 

Public Class Form1 

Dim MyDB As New NorthWind.NorthWind("Northwind.sdf") 

 

 طلبنا منه أن ينشئ نطاق sqlmetal وذلك لأننا عند انشاء الفئات NorthWind.NorthWindطبعا تلاحظ انا كتبنا 
  . فلابد للوصول لفئاتنا استخدام ھذا النطاق قبلھا NorthWind باسم Namespaceللأسماء 

 أكثر ونقول LINQالان اصبحت كل البيانات في متناول يدك وسنبدأ بتجربة اكواد مختلفة لنتعرف على استعلامات 
  . الممل والمكرر ADO.Net وكود SQLتعلامات سوداعا لا

  

  LINQاستعلامات 
 وتعيد عندھا قيمة Aggregate وتعيد مجموعة سجلات أو ب From اما تبدأ ب LINQبشكل عام استعلامات 

  . المقابل له ليسھل فھمه SQLواحدة وساقوم بترجمة كل استعلام نستخدمه لاستعلام 

 :Fromاستعلامات : اولا  

  صيغتھا العامة المبسطة

From element In collection [Where condition] Select elements[vars] 

 التي تمثل قاعدة البينات كاملة وطبعا ما بين اقواس مربعة ھو MyDB تصبح Collectionوفي مثالنا تصبح 
  .اختياري 

   .DataGridView لتعرض البيانات في Form_Load بسيطة في الحدث fromالان سنضع جملة 
 
Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As 
System.EventArgs) Handles MyBase.Load 
 
       MyDB.Log = Console.Out 
 
       Dim mySource = From cust In MyDB.Customers Select cust 
 
       DataGridView1.DataSource = mySource 
 
End Sub 
 

 

 MyDB.Log = Console.Outولقد اضفت العبارة .  كاملا لديك Customersنفذ البرنامج وسترى الجدول 
 المعروفة فھذه ھي SQL ما يتم ارساله لقاعدة البينات وسترى انھا استعلامات Outputحيث ستكتب لك في نافذة 

  .بعد كل استعلام Output ترجمة ما نكتبه لقاعدة البيانات ولاتنسى ان تلقي نظرة على نافذة LINQ to SQLوظيفة 

 custلكل سجل نفترض ان اسمه "فترجمته لعبارة مفھومة ھو ،، سجل custأعود للاستعلام السابق بامكانك اعتبار 
   :SQLيعني كل السجلات ومقابله ب  " cust اعطني MyDB.Customersفي مجموعة السجلات 

SELECT * FROM Customers  

  

ات  ولاتشغل بالك كثيرا به فالتطبيقAnnonymous Type فھو نوع مجھول mySourceاما ناتج الاستعلام 
يمكنك اعتبار الناتج مجموعة تحتوي كائنات من نوع . لاتحتاج لمعرفة قوية بھذه الانواع LINQالبسيطة ل 

Customer.  

أو يمكنك  . Data Sourceوناتج الاستعلام يمكنك الاستفادة منه بعدة طرق مثلا في المثال السابق استخدمناه ك 
  : مثال For..Eachالتنقل خلاله من خلال حلقة 

  

For Each mycust In mySource 
 
Console.WriteLine(mycust.ContactName) 
 
Next 

  

  ...أو اخذ اول سجل منه فقط واستعراض بيناته مثلا 

  

Dim FirstCust = mySource.First 
 
Console.WriteLine(FirstCust.CompanyName, FirstCust.ContactName, 
FirstCust.City, FirstCust.Address) 

  مثلا .. جديدة Fromوفوق ذلك يمكن الاستعلام عنه في جملة ..  ينطبق عليه List(of T) ل ما ينطبق علىوك

  

Dim LondonCust = From x In mySource Where x.City = "London" Select x 

  . للداتا جريد DataSource كمصدر LondonCustوبامكانك ان تضع ناتج الاستعلام 

  

 يقترح عليك العناصر و الحقول ويصحح لك الأخطاء Visual BASIC للاستعلامات السابقة أن لاحظ اثناء كتابتك
 Linq فلامجال مع SQL اسھل واصح من استخدام استعلامات LINQوھذا سبب قولي أن استخدام استعلامات 

 لنوع بيانات خاطئ مثلا ويكون سببه بسيط كاختيارك.لخطأ لاتعرف ما ھيته ترسله لك قاعدة البينات وتحتار في امره 
String بدلا من Integerفھنا المحرر سيخبرك بالخطا مباشرة .  

  

 وشاھد النتيجة وبعده Form_Load وقم انت باستبداله في حدث LINQنعود لبعض التطبيقات وساسرد ھنا استعلام 
  ._  في سطر باستخدام  محجوزة المقابل ولسھولة فھم الاستعلامات ساكتب كل بداية كلمة SQLساذكر استعلام 

 

Dim mySource = From cust In MyDB.Customers _ 
               Select cust.ContactName, cust.City 
 
SQL : SELECT ContactName,City FROM Customers 
 
 
Dim mySource = From cust In MyDB.Customers _ 
              Select cust.ContactName, cust.City 
 
SQL : SELECT ContactName,City FROM Customers 
 
 
 
Dim mySource = From cust In MyDB.Customers _ 
               Where cust.City = "London" 
              Select cust.ContactName, cust.ContactTitle, cust.Address 
SQL : SELECT ContactName, ContactTitle, Address FROM Custemers WHERE 
City='London' 
 
 
 
 
Dim mySource = From ord In MyDB.Orders _ 
             Where ord.Customer.City = "London" _ 
             Select ord.OrderDate,ord.OrderID,ord.Customer.ContactTitle 
 
SQL : SELECT Orders.[Order Date],Orders.[Order ID],    
Customers.[Contact Title] FROM Orders 
INNER JOIN Customers ON Customers.[Customer ID] = Orders.[Customer ID] 
WHERE Customers.City = 'London' 

  

 Customer يحتوي على كائن Orderلاحظ الاستعلام الاخير أن العلاقات ممثلة بشكل مجموعة كائنات فكل كائن 
 في الجدولين وھذه طريقة ذكية وسھلة لتمثيل العلاقات خاصة العلاقات المتشابكة Customer IDة بين يمثل العلاق

وطبعا يمكنك استخدام الكلمات .  في الاستعلام Inner Joinبين عدة جداول حيث كنت سابقا اكتب اكثر من جملة 

عليك ان تجرب ما يقترحه عليك المحرر  واقترح Distinct ,Order By ,Likeالمحجوزة التالية مع الاستعلامات 
حيث ساتطرق ل .الى لقاء قريب .  اكثر تعقيدا لتعتاد على استخدامھا LINQوتحاول بناء جمل استعلامات

Aggregate و استعلامات اكثر تعقيدا .  

 ٨

  :اليوم الثالث 
  

   ولن ادخل بكل تفاصيل الكلمات المحجوزة للاستعلامات LINQاليوم سنكمل مشوارنا مع استعلامات 
  

   Aggregateاستعلامات : ثانيا 
  

   Aggregationتتميز بانھا تعيد قيمة واحدة لأحد دوال التجميع 
Sum مجموع القيم – Average المعدل الحسابي – Count عدد القيم – Min أصغر قيمة – Max أكبر قيمة .  

  وھي سھلة الاستخدام 
  ...ال أضف الكود التالي لبرنامجنا في أي مكان تريد مث

 
Dim CustomerCount = Aggregate cust In MyDB.Customers Into Count() 
'SQL : SELECT Count(*) From Customers 
Dim OrderSum = Aggregate ord In MyDB.Orders Into Max(ord.OrderDate) 
'SQL : SELECT Max([Order Date]) FROm Orders 
Console.WriteLine("Custuomers Count: {0} Type : {1} ", CustomerCount, 
TypeName(CustomerCount)) 
Console.WriteLine("Max Order Date Count: {0} Type : {1} ", OrderSum, 
TypeName(OrderSum)) 

  
  

 تعيد قيمة واحدة دائما ونوع Aggregateعلامات  الموافقة بعد الاستعلام وكما تلاحظ فإن استSQLوقد اضفت جملة 
كما يمكن ان يحتوي . كما اعتدنا Collection وليس مجموعة DataTypeمتغير ناتج الاستعلام سيكون أحد انواع 

   Aggregate على تجميع باستخدام Fromاستعلام 
  

   :LINQ TO SQLالعلاقات في 

 اصبح لدينا تطبيق سھل ووصول جيد لكافة كائنات قاعدة Linqتخدام ومع اس. طبعا جميعنا يعرف العلاقات وفائدتھا 
 ولذلك قبل الخوض في علاقاتھا ھذه صورة توضح NorthWindوكما سبق فإننا نستخدم قاعدة بيانات ... البيانات 

  بنية العلاقات فيھا
  

  
  
  

 يسھل بناء sqlMetalباستخدام طبعا كما ذكرت انشاء العلاقات وحفظھا في قاعدة البيانات قبل انشاء الفئات 
بتضيمن العلاقات وتقديمھا لنا  ) O/R designerأو  sqlmetal(استعلاماتنا بشكل كبير حيث يقوم باني الفئات 

مثلا من الشكل السابق لنفرض أنني اريد معرفة فقط اسماء وارقام المنتجات التي شحنت . بطريقة برمجية ككائنات 
  : بھذه البساطة LINQ استعلام سيكون عندھا. الى ايطاليا 

  
From det In MyDB.OrderDetails _ 
Where det.Order.ShipCountry = "Italy" _ 
Select det.Product.ProductID, det.Product.ProductName Distinct 

  
 ويصبح  Join بحاجة ل أما اذا لم يتم انشاء العلاقة في قاعدة البيانات فانت. ھذا في حالة ترجمة العلاقات في الفئات 

  :الاستعلام نفسه كالتالي 
From det In MyDB.OrderDetails _ 
Join ord In MyDB.Orders On ord.OrderID Equals det.OrderID _ 
Join prd In MyDB.Products On prd.ProductID Equals det.ProductID _ 
Where ord.ShipCountry = "Italy" _ 
Select prd.ProductID, prd.ProductName Distinct 

  
  : ھو SQLوطبعا نظير الاستعلامين السابقين في 

SELECT DISTINCT Products.[Product Name] FROM [Order Details] 
INNER JOIN Orders ON [Order Details].[Order ID] = Orders.[Order ID] 
INNER JOIN Products ON [Order Details].[Product ID] = Products.[Product 
ID] WHERE   (Orders.[Ship Country] = 'Italy' 

  
-Primary(  حيث يكون ھناك جدول مفتاح اساسي Foreign Keyوالنتيجة أن علاقة واحد إلى كل أو ما تسمى 

Key  (  مثل جدولCustomers و جدول اخر يحتوي المفتاح الفرعي )Foreign-Key  (  الذي يشير للمفتاح
  : بشكل Linq to SQLسي  تكون ممثلة في فئات الاسا

   )Foreign-Key له مجموعة من كائنات جدول Primary-Keyكل كائن من جدول ( 
   والعكس غير صحيح 

   )Primary-Key ممثل بكائن وحيد من جدول Foreign-Keyكل كائن من جدول (
  .بقوھذه الجمل صحيحة مھما تشعبت العلاقات كما شاھدنا بالمثال السا

وھذه الميزة لھا عيوب عند التعامل مع كمية كبيرة من البيانات حيث سيتم تحميل الجدول المطلوب مع علاقاته 
 للكائن Load Optionsوالكائنات الفرعية مما يشكل حملا على الذاكرة والمعالج وھذا يمكن التحكم به من خلال 

DataContext كما سنرى لاحقا .  

  :لاستعلامات الاستخدامات الاساسية ل
 :Data Sourceبناء مصدر بيانات   

 كمصدر بيانات لأي عنصر تحكم لعرض البيانات Linqكما رأينا سابقا في المقال الثاني يمكن استخدام أي استعلام 
 وھي أن أي مجموعة استعلامات تعيد نفس القيمة أو السجل أو نفس Linqوھناك ملاحظة مھمة بالنسبة لاستعلامات 

فقد تتفاجأ عند وضع مصدر بيانات ل . لسجلات تؤشر لنفس الكائن حتى لو كانا استعلامين مختلفين مجموعة ا
combobox و DataGrid من نفس الجدول أنه عند تحريك الاختيار في الكومبو تتحرك المؤشرات في datagrid.  

 :Groupingالتجميع   
لنفرض أنني لا أريد .ت مستقلة وفقا لمعيار معين  حيث يتم فرز السجلات على شكل مجموعاGroup Byباستخدام 

   group By عندھا نلجأ ل Category بالكامل وانما اريده بطريقة منسقة ومجمعة حسب Productsعرض جدول 
From pro In MyDB.Products _ 
Group pro By pro.CategoryID Into CatGroup = Group 

 للتجوال عبر For Eachتين بشكل شجري بالتالي نحن بحاجة لحلقتين ھنا يصبح ناتج الاستعلام عبارة عن مجموع
  :  والحلقة الثانية تتجول في ھذه المجموعة كالتالي CatGroupجميع السجلات الحلقة الأولى الأولى تعيد مجموعة 

         
 
Dim mySource = From pro In MyDB.Products _ 
               Group pro By pro.CategoryID Into CatGroup = Group 

For Each grp In mySource 
     Console.WriteLine("Category ID = {0}", grp.CategoryID) 
     For Each prod In grp.CatGroup 
           Console.WriteLine("Product Name: " & prod.ProductName) 
     Next 
Next 

  
 لكل ordersعا للتجميع ھو مع الدوال التجميعية  مثلا اذا اردنا أن نجد عدد الطلبيات والاستخدام الاخر الأكثر شيو

  : سنستخدم الاستعلام التالي Employeeموظف 
  

From ord In MyDB.Orders _ 
Group ord By ord.EmployeeID Into EmpGroup = Group _ 
Select EmployeeID, OrdersCount = EmpGroup.Count() 

  
  :تيب التر

 ١٠معھا لتعطينا جزء من المجموعة مثلا للحصول على اخر  Skip و Take  ويمكن استخدام Order Byباستخدام 
  :طلبيات نستخدم الاستعلام 

  
From ord In MyDB.Orders _ 
Select ord Order By ord.OrderDate Descending _ 
Take 10 

  
:تصفية السجلات   

 Linq toمن القيم المنطقية والدوال مقبولة بالنسبة % ٩٠عدھا  قيمة منطقية و وبWhereوھو باستخدام الكلمة 
SQL ولاحقا سأعطي أمثلة للدوال واستخدامھا بطريقة سھلة مع Linq.   

  
لكي تأخذ مرونة في . جرب ثم جرب ثم جرب كل صيغ الاستعلامات الممكنة LINQونصيحتي لكل متعلم جديد ل 

  ....التعامل 
   .Linqع المقال القادم الذي سيكون عن تعديل البيانات واضافة وحذف البيانات مع وإلى اللقاء م

  ....والسلام 
  
  
  
  
  
  
  
  
  
  
  
  
  
  

اليوم الرابع   

  تعديل وتحديث البيانات
  السلام عليكم

 أصبحنا نتعامل مع كائنات وبالتالي Linqكمدخل لمفاھيم تعديل وتحديث البيانات أحب أن اذكر الجميع بأننا في 
أو حذف أو اضافة كائنات ثم يقوم مفھوم تحديث البيانات ھو عبارة عن تعديل لخصائص ھذه الكائنات والمجموعات 

  . يتم ارسالھا لقاعدة البيانات SQLتحويل تعديلاتنا إلى جمل  بLinq to SQLمزود 
  

  :أولا إضافة البيانات 
  
  : إضافة البيانات لجدول واحد -١

 نسند القيم للحقول المطلوبة ثم  ) Category أو Customerمثلا ( نقوم أولا بانشاء نسخة من كائن الجدول 
 يقوم لوحده بتحديث بقية LINQاعدة البيانات والميزة الجميلة أن  لإلحاق البيانات بقDataContextونستخدم الكائن 

 NorthWindفي Categoriesالفئات مثلا الكود التالي يضيف بيانات  لجدول . الحقول التي لم ندخلھا لقيمھا الجديدة
 . الجديد للفئة IDثم يعرض قيمة 

    Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As 
    System.EventArgs) Handles Button2.Click 
        Dim MyDB As New NorthWind.NorthWind("Northwind.sdf") 
        MyDB.Log = Console.Out 
        Dim myCat As New NorthWind.Category 
        With myCat 
            .CategoryName = "Arabicaa" 
            .Description = "Arabs" 
        End With 
        MyDB.Categories.InsertOnSubmit(myCat) 
 
        MyDB.SubmitChanges() 
        MsgBox("new ID : " & myCat.CategoryID) 
 
    End Sub 

 
   .Outputلاتنسى أن تلقي نظرة على نافذة 

  :Details/Masterإضافة البيانات لجداول مترابطة -٢
غالبا ما نحتاج ادخال المستخدم لإضافة البيانات لعدة جداول مرتبطة مثلا عند ادخال العميل لفاتورة ما نحتاج لإدخال 

 لنفرض Northwindوفي قاعدة ... التفاصيل البيانات في جدول الفواتير ثم نضيف تقاصيل كل فاتورة في جدول 
 Categoryما سنقوم به ھو انشاء كائن جديد من نوع .. جديدة تتبع لفئة جديدة Productsأننا نريد إدخال منتجات 

 فنضيف لھا ما نحتاجه من منتجات و نرسل التغييرات لقاعدة Productsوسنجد في خصائصه تلقائيا مجموعة 
 :الكود وھذا مثال .البيانات 

Dim myCat As New NorthWind.Category 
With myCat 
.CategoryName = "Arabicooo" 
.Description = "Arabians" 
.Products.Add(New NorthWind.Product With _ 
             {.ProductName = "AAAAAA",.Discontinued = False}) 
.Products.Add(New NorthWind.Product With _ 
             {.ProductName = "BBBBBB",.Discontinued = False}) 
.Products.Add(New NorthWind.Product With _ 
            {.ProductName = "CCCCCC",.Discontinued = False}) 
End With 
 
MyDB.Categories.InsertOnSubmit(myCat) 
 
MyDB.SubmitChanges() 
MsgBox("CategoryID ID : " & myCat.CategoryID & " My Products : " & _  
         myCat.Products.Count) 



واذا القيت نظرة على نافذة .  سيتكفل باحضارھا Linq to SQL في كودنا لأن CategoryIDولاحظ أننا لم نضع 
Output ستجد أن كودنا السابق تمت ترجمته لخمسة جمل SQL الأولى لإضافة فئة جديدة Category والثاني 

  . الجديدة Productجات  للفئة الجديدة و الثلاث البقية لإضافة المنتCategoryIDللحصول على 
  

  :ثانيا تعديل البيانات 
  

مثلا ) أرسله ...عدله ..أحضر الكائن المطلوب . ( اذا فھمت الفقرة السابقة فالتالي أسھل باختصار  طريقة التعديل 
   ھذا الكودXXXXX ليصبح اسمه CCCCCCلنعدل أحد المنتج السابق الذي أضفناه المسمى 

  
Dim myProduct = From pro In MyDB.Products _ 
                Where pro.ProductName = "CCCCCC" Select pro 
If myProduct.Count > 0 Then myProduct.First.ProductName = "XXXXXX" 
MyDB.SubmitChanges() 

 
  

  :ثالثا حذف البيانات 
  

وھذا كود لحذف ...)  قاعدة البيانات طبعا (أحضر الكائن ثم احذفه ثم أخبر القاعدة :  ايضا بنفس الطريقة السابقة 
   :XXXXXالمنتج 

Dim myProduct = From pro In MyDB.Products _ 
              Where pro.ProductName = "XXXXXX" Select pro 
If myProduct.Count > 0 Then _ 
  myDB.Products.DeleteOnSubmit(myProduct.First) 
MyDB.SubmitChanges() 

 
 Arab التي تبدأ ب Category حيث سنحذف Northwindولحذف كل ما اضفناه في الأكواد السابقة لقاعدة بيانات 

  :ومنتجاتھا  بھذا الكود 
  

 
 
 
 
 
 
Dim myCat = From cat In MyDB.Categories _ 
            Where cat.CategoryName Like "Arab*" Select cat 
 
If myCat.Count > 0 Then 
      For Each cat In myCat 
        MyDB.Products.DeleteAllOnSubmit(cat.Products) 

Next 
End If 
 
MyDB.Categories.DeleteAllOnSubmit(myCat) 
MyDB.SubmitChanges() 
 
 
 

  
  
  
  



   :sactionsTranنطاق العمليات :رابعا 
  

 خاصة عند حدوث خطأ Transactionاعتدنا دائما لإنشاء برنامج يعتمد عليه أن نستخدم عن تغيير البيانات كائنات 
فليس من المعقول أن نسجل الفاتورة مثلا ناقصة بعض ..ما أثناء ادخال بيانات مترابطة فيجب الغاء العملية برمتھا 

  .الاغراض 
  

 سيقوم مزود SubmitChanges  داخليا فعند أي استدعاء ل Transactionھي تدعم  فLinq To SQLوبالنسبة ل 
Linq تلقائيا بانشاء Transaction خاص بھذا الاستدعاء حيث ولدى حدوث أي خطأ فإن عملية تحديث البيانات 

ن أولد خطأ عن حيث تعمدت أ. (  المقصودة لازالت موجودة Categoryجرب ھذا الكود أن .. سيتم الغائھا تلقائيا 
 وھذا لا تسمح به قاعدة البيانات حيث يجب حذف Product مرتبط بكائنات من Categoryطريق حذف كائن من 

  ).الطرف الآخر من العلاقة 
  

Dim myCat = From cat In MyDB.Categories Where cat.CategoryID = 2 
 
MyDB.Categories.DeleteAllOnSubmit(myCat) 
 
Try 

MyDB.SubmitChanges() 
Catch ex As Exception 

MsgBox("Error !! all changes canceled") 
End Try 
 

  . مرة أخرىSubmitChanges ثم نستدعي Catchويمكن تعديل الكود السابق بحيث نتتبع الخطأ و نعالجه في جملة 
  

ا استخدام نطاق  حيث بامكاننDataContext للكائن Transactionوھناك طريقة أخرى عن طريق خاصية 
Transaction من كائنات ADO.net وستكون كل العمليات التي ستتم ضمن ھذا النطاق واذا عملنا RollBack 

  .ستلغى كل التغييرات 
  

إلى لقاء مع الموضوع القادم الذي سيكون عن ميزة رائعة يمكن .. امل أن يكون مفيدا للجميع ..بھذا ينتھي مقال اليوم 
   .DataContextوايضا سنلقي نظرة أعمق على الكائن  ..Extensions وھي Linqي الاستفادة منھا ف

  
  ...والسلام 

اليوم الخامس   
  Extensionsتوسعة الوظائف 

  
ن ونح ) String,Integer,List(  له مجموعة من الدوال كما اعتدنا مثلا VB2008إن كل نوع بيانات أو فئة في 

ولكن ماذا لو احتجنا لتوسيع ھذا النوع ليحتوي وظائف جديدة .. نستخدم ھذه الوظائف بشكل اعتيادي في كل برامجنا 
سابقا كنا ننجز ھذا العمل باشتقاق فئة جديدة و اضافة الوظائف لھا ولكن الآن يمكننا بسھولة .. نحتاجھا في البرنامج ؟ 

   . Extensionsدام إضافة وظيفة أو أكثر لنوع بيانات باستخ
لنفترض أننا نريد الحصول ..  ھذه الميزة أحيانا تغنينا عن كتابة أكواد كثيرة ومتكررة لتحقيق ھدف معين Linqفي 

 تحقق الھدف المطلوب ولذلك SQL أو في Linqلن نجد دالة تجمع في ... على مجموع حقول نصية لعدة سجلات 
جزء ) …Sum,Count(ل الذي نريد تعديله لتصبح كبقية دوال التجميع سننشيء دالتنا الخاصة ونلحقھا بنوع الحق

  .من نوعنا 
 تقوم بعرض مجموعة حقول JoinNamesنسميھا ) وظيفة (مھمتنا الحالية ھي انشاء دالة : ترجمة للكلام السابق 

ات  عرض الفئNorthWindوسيكون مثالي من قاعدة بيانات . نصية بشكل حقل نصي واحد مفصول بفواصل 
Categories بحيث يكون بجانب كل سجل من Category مجموعة المنتجات Products التي تخص ھذه الفئة 

  .مفصولة بفواصل 
 يعني بلغة البرمجة               Product وبالأصح مجموعة Productفالنوع المطلوب توسعته ھنا ھو 

IEnumerable( of Product)   
  :الكود كالتالي بكل بساطة 

  
 
Imports System.Runtime.CompilerServices 
… 
… 
… 
Public Module Ext 
    <Extension()> Public Function JoinNames(ByVal ProductList As 
IEnumerable(Of NorthWind.Product)) As String 
 
        Dim Names = From pro In ProductList Select pro.EnglishName 
 
        Dim NamesString = String.Join(",", Names.ToArray) 
        Return NamesString 
    End Function 
 
End Module 

  
  

   :datagridالان لنستخدم ھذه الدالة الجديدة في مثالنا و نعرضھا في 
  

 
Dim mySource = From cat In MyDB.Categories _ 
               Let ProductsNames = cat.Products.JoinNames _ 
               Order By cat.CategoryName _ 
               Select Number = cat.CategoryID,Name =cat.CategoryName, _ 
               Products = ProductsNames 
 
DataGridView1.DataSource = mySource 

   DataContextنظرةأعمق على 

  
   :Linq to SQL و تفردھا مع خصوصية الكائنات

  
 مع قواعد البيانات فھو الذي يترجم ما Linq To SQL كما ذكرنا سابقا ھو وسيط اتصال DataContextان الكائن 

  !!كيف ذلك !! وايضا من مھماته  التحكم باحضار البيانات والتحقق منھا ؟؟ . SQLنعمله بالكائنات إلى جمل 
المعنى أنه لو ...  لايكرر وجود نفس البيانات DataContextسباب أخرى فإن لكي لايحدث تضارب بالبيانات ولأ

استدعيت نفس مجموعة السجلات في مكانين مختلفين وبطريقتين مختلفتين فسيكون لديك في الذاكرة نفس المجموعة 
  :وھذا مثال !! من السجلات 

  
Private Sub Button1_Click(ByVal sender As System.Object, ByVal e As 
System.EventArgs) Handles Button1.Click 
 
''Ado.net Objects        
Dim con As New SqlServerCe.SqlCeConnection("Data Source = 
Northwind.sdf") 
Dim com1 = con.CreateCommand 
com1.CommandText = "Select Top (1) * From Categories Where [Category 
ID] = 3" 
con.Open() 
Dim Cat1 = com1.ExecuteResultSet(SqlServerCe.ResultSetOptions.None) 
Dim Cat2 = com1.ExecuteResultSet(SqlServerCe.ResultSetOptions.None) 
con.Close() 
''LINQ to SQL DataContext 
Dim MyDB As New NorthWind.NorthWind("Northwind.sdf") 
MyDB.Log = Console.Out 
Dim myCat = From catg In MyDB.Categories _ 
            Where catg.CategoryID = 3 Select catg 
Dim Cat3 = myCat.First 
Dim Cat4=MyDB.Categories.Where(Function(cat) cat.CategoryID = 3).First 
 
 التساوي فحص ''
Console.WriteLine("Cat1,Cat2 are Equals ? " & Cat1.Equals(Cat2)) 
Console.WriteLine("Cat3,Cat4 are Equals ? " & Cat3.Equals(Cat4)) 
 
End Sub 

  
 أن يكون الكائتنين مختلفين رغم أنھما متساويين بالمحتويات جرب الكود السابق وسترى ADO,Netنحن تعودنا في 

  )  ولكنه قابل للكتابة DataReader مثل ResultSetللتذكرة كائن (  متساويين resultsetأن كائني 
ما أھمية ذلك ؟؟؟ من ذلك نسنتنج أن تنتبه عند أجرائك تعديلات على الكائنات فحتى الكائنات التي كنت تعتقد أنھا 

   .datacontextتحتفظ بالقيم الأصلية قبل التعديل ستكون قد تعرضت للتغير من قبل 
   Linq to SQLما الحل ؟؟  خذ ھذ القاعدة و طبقھا دائما في 

  
  :ملاحظة  ھامة 

 وتصنع كائنا واحدا لكل Connection لاستعلاماتك يعني لاتعامله مثل DataContextدائما انشئ كائن جديد 
 كما في جميع امثلتي السابقة اذا Procedure Levelالتطبيق انما اجعله كائنا على مستوى الاجراء أو الوظيفة 

  .لاحظتم 
    

  
  
  

  :يانات قاعدة البالتحميل من 
  

 فإننا لانقوم فعليا الا بتعريف الاستعلام وقولبة المتغيرات Linq to SQLعندما نكتب استعلاما ما من استعلامات 
لعرض النتائج )  وارسالھا لقاعدة البيانات SQLأي ترجمته لجملة ( ويتم تنفيذ الاستعلام . لاستقبال نتائج الاستعلام 

   اذا كنت وضعت السطرOutPutذلك بمشاھدة نتائج نافذة عند الحاجة لذلك ويمكنك ان تلاحظ 
  

MyDB.Log = Console.Out 
  

  مثلا اذا كتبنا السطر التالي 
  

Dim myCat = From catg In MyDB.Categories Select catg 
 

  For Eachفھو عبارة عن تعريف الاستعلام وتنفيذه الفعلي يتم عندما يحتاج البرنامج للبيانات وذلك أما بحلقة 
 جرب كتابة كود لذلك وتتبع تنفيذ البرنامج بوضع datasource أو عند اسناده ل ToListأو عند استدعاء الأمر 

BreakPoint و سترى متى يتم عرض جملة SQL .   
  

  فإذا أردنا أن يتم التنفيذ فورا فقط حول السطر السابق إلى ما يلي 
 
Dim myCat = (From catg In MyDB.Categories Select catg).Tolist 

 
   .SubmitChanges ھي عند اعتماد التغييرات باستخدام SQLوالحالة الأخرى التي يتم فيھا ارسال جملة 

  
  :فھناك عدة خيارات  ) أي المرتبطة بالعلاقات ( وأما بالنسبة للكائنات الفرعية 

  
 ObjectrackingEnabledا تكون خاصية  وھو الخيار الافتراضي وھن:تحميل الكائنات الفرعية عند الحاجة لذلك 

  .عند اللزوم ) جدول مرتبط ( وھنا يتم تحميل أي كائن فرعي Trueمعينة للقيمة 
   عندھا يتم تحميل الكائنات الأساسية فقط ولايتم تحميل الكائنات الفرعية  وذلك :تحميل الكائنات الأساسية فقط 

  !! .وتصبح البيانات للقراءة فقط . ObjectrackingEnabled للخاصية Falseباسناد القيمة 
 فنتحكم نحن بما يتم تحميله من قاعدة Load Options وذلك عن طريق خيارات التحميل: التحكم بتحميل الكائنات 

  أن يحمل مع كل  Orders أنه عند تحميل مجموعة  DataContextمثلا بامكاننا أن نخبر . البيانات مع كل كائن 
  ObjectTrackingEnabled يرجع لخاصية DataContext و ھنا فإن OrdersDetailsعة  مجموOrderكائن 

  .لمعرفة ما إذا كان يجب تحميل بقية الكائنات الفرعية أم لا 
وقبل ذلك فإن طريقة استخدام . لتوضيح الأمر أكثر سأناقش ھذه الحالات  لاحقا ثم أعرض مثال توضيحي

LoadOptions كالتالي  :  
    :OrdersDetails مع كائن Ordersني أريد تطبيق المثال في الأسطر السابقة أي تحميل كل كائن لنفترض أن

  
Dim MyDB As New NorthWind.NorthWind("Northwind.sdf") 
MyDB.Log = Console.Out 
 LoadOptionsانشاء آائن ''
Dim myopt As New System.Data.Linq.DataLoadOptions 
''Order للكائن نريد تحميله من العلاقات ماتحديد  
myopt.LoadWith(Of NorthWind.Order)(Function(o As NorthWind.Order)o.OrderDetails) 

'' DataContext ربط الخيارات السابقة مع 
MyDB.LoadOptions = myopt 

  
 يطلب الكائنات التي DataContext لنترك كل شي كما ھو افتراضيا وندع قد يتسائل شخص ما ما فائدة ذلك ؟؟؟

  !!نحتاجھا 
 فسيكون الناتج لدينا حوالي OrderDetailsو  Order افترض أن لديك دالة تكرارية تستدعي كائنات : وجوابي ھو 

 Orders مباشرة مع الكائن OrderDetails سيرسل لقاعدة البيانات بينما بامكانك تحديد تحميل sql استعلام ١٠٠
كثيرون يشتكون من أداء .(ياناتك سوى استعلام واحد جرب ذلك ولاحظ الفرق في الأداءوعندھا لن يرسل لقاعدة ب

Linq  والسبب ھو أن كودھم يواصل ارسال الاستعلامات ويقتل الأداء .  
  

  ....الآن لنوضح كل ما ذكرناه سابقا  بمثال عملي 
  OutPut ولكن تتبع تنفيذه و راقب نافذة ومثالي ھذا الكود وھو لايعمل شيئا سوى تحميل كائنات من قاعدة البيانات

  :لترى مايحدث توضيح الخطوات موجود مع الكود 
  

Private Sub Button2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) 
Handles Button2.Click 
 
  1 تجربة ''''''''''''''''        
        Dim MyDB1 = CreateDataContext() 
 الطب عند الكائنات تحميل تمكين''        
        MyDB1.ObjectTrackingEnabled = True 
        ''First لوجود مباشرة الاستعلام تنفيذ سيتم 
        ''OutPut نافذة من البيانات لقاعدة إرساله يتم الذي الاستعلام الخطوة في راقب هنا  
        Dim Q1 = (From ord In MyDB1.Categories Select ord).First 
 ترى آما اخر استعلام تنفيذ تم هنا''        
        Dim Q2 = (From prod In Q1.Products Select prod).ToList 
 آائنات وجود النتيجة ''        
        MsgBox(Q2.Count) 
 
 2 تجربة ''''''''''''''''''        
        Dim MyDB2 = CreateDataContext() 
 الكائنات تحميل الغاء ''        
        MyDB2.ObjectTrackingEnabled = False 
 السابقة الاستعلام جملة نفس لاحظ ''        
        Dim Q3 = (From ord In MyDB2.Categories Select ord).First 
  شيء أي ارسال لايتم هنا لاحظ ''        
        Dim Q4 = (From prod In Q3.Products Select prod).ToList 
 خالية النتيجة ''        
        MsgBox(Q4.Count) 
 
        Dim myOpt As New System.Data.Linq.DataLoadOptions 
        ''Products الفرعية الكائنات تحميل ذآرناه آما تعريف  
        ''Category للكائن  
        myOpt.LoadWith(Of NorthWind.Category)(Function(cat As NorthWind.Category) 
cat.Products) 
 
 3 تجربة ''''''''''''''''        
        Dim MyDB3 = CreateDataContext() 
 الطلب عند التحميل تمكين''        
        MyDB3.ObjectTrackingEnabled = True 
 التحميل خيارات تحديد ''        
        MyDB3.LoadOptions = myOpt 
 اختلف آيف الاستعلام راقب الآن ''        
        Dim Q5 = (From ord In MyDB3.Categories Select ord).First 
 تحميلها سبق لأنه هنا استعلام لايوجد ''        
        Dim Q6 = (From prod In Q5.Products Select prod).ToList 
 نتائج يوجد ''        
        MsgBox(Q6.Count) 
 
 4 تجربة ''''''''''''''''        
        Dim MyDB4 = CreateDataContext() 
 الطب عند التحميل الغاء ''        
        MyDB4.ObjectTrackingEnabled = False 
 التحميل خيارات ''        

        MyDB4.LoadOptions = myOpt 
 ايضا الاستعلام جملة لاحظ ''        
        Dim Q7 = (From ord In MyDB4.Categories Select ord).First 
 ينفذ لااستعلام ''        
        Dim Q8 = (From prod In Q7.Products Select prod).ToList 
 موجود النتائج ايضا ''        
        MsgBox(Q8.Count) 
 
    End Sub 
    Function CreateDataContext() As NorthWind.NorthWind 
        '' DataConteXtلإنشاء بسيطة دالة  
        Dim myDB As New NorthWind.NorthWind("Northwind.sdf") 
        myDB.Log = Console.Out 
        Return myDB 
    End Function  

  
  

 ليضع قدمه على LINQ to SQLنني استطعت على عجالة إعطاء دفعة لمن يريد الانتقال لاستخدام ھكذا أعتقد أ
  . أرجو أن أكون قد وقفت في العرض والأمثلة . طريق البداية

" ھل من سؤال؟"مع أني لاحظت عدم وجود أي أسئلة أو استفسارات وقد كان أحد أساتذتي عند كل نھاية درس يسأل 
  "أو ولا واحد كان سامعني ... ..إما كلكم فاھمين "سأل يقولوعندما لا أحد ي

ولم نتنھي بعد حيث لدي أيضا بعض الأفكار و التقنيات لاستخدام .المھم أن يكون طرحي مفيدا ولاتنسونا من الدعاء 
Linq to SQL عليكموالسلام .....  وسأطرحھا على شكل مواضيع مستقلة  

